
  1: Module  CHERST1
  2: Title   'MUX DATA FROM HP TO ASIC for CHER board, second part'
  3:
  4: "------------------------------------------------------------------"
  5: " This CPLD is used by CHERII and contains the following logic:    "
  6: " 1)The Header block is checked for size and a strobe is generated "
  7: "   on the 16th word for a pedestal clock [PEDCLK]                 "
  8: " 2)The Header bad latch is here, output is [HBAD#]                "
  9: " 3)The Busy error latch is here, output is [BERR#]                "
 10: " 4)End of Event is a level generated output [EOFEVNT]             "
 11: " 5)A signal is generated to put header data from the low half word"
 12: "   on the high half for pedestal use, output is [XDATA]           "
 13: " 6)A pulse is generated [CLRWDT#] to reset the event arrival WDT  "
 14: "   on event end or FE_CLR# or PRWRST#                             "
 15: " 7)A second pulse is generated[EECLR#] to reset the event process "
 16: "    WDT on FE_CLR# or PRWRST#.                                    " 
 17: " 8)The signal [HEADCNT] is generated to latch the header data     "
 18: "                                                                  "
 19: " PATH: stardaq/home2/bob/powerview/cher1 cherst1.abl              "
 20: " Fix xdata     6/2/98       cherst1.abl -I jedec file             "
 21: "------------------------------------------------------------------"
 22:
 23: " Inputs
 24: "HPSTRB  PIN;"
 25: ADJSTRB  PIN;
 26: !HPCAVT  PIN;
 27: !HPDAVT  PIN;
 28: HPDA00  PIN;
 29: HPDA01  PIN;
 30: HPDA02  PIN;
 31: "HPDA03  PIN;"
 32: FE_CLR  PIN;
 33: ST00  PIN;
 34: ST01   PIN;
 35: BUSY    PIN;"Input to check for data overrun and set busy fault."
 36: ABORTP  PIN;
 37: !CLRBSY0  PIN;
 38: !CLRBSY1  PIN;
 39: !CLRBSY2  PIN;
 40: !LINKRDY  PIN;
 41: TPCMODE  PIN;"Enable Pedestal strobe for TPC mode"
 42: "!MRST   PIN_29; used for powerup reset, this gets 'ored' with FE_CLR"
 43: "             and ABORT at the GLB level"
 44: PLSI PROPERTY 'Y1_as_reset on';
 45:                
 46: " Outputs
 47: HQ6..HQ0 node istype 'reg_T';
 48: ST01D    node istype 'reg_D';
 49: SEREN    node istype 'reg_SR';
 50: PEDCLK   pin istype 'reg,buffer';
 51: MRESET   node istype 'com';
 52: HEADCNT  pin  istype 'com,buffer';
 53: HEADCLK  node istype 'com';
 54: HEADB15  node istype 'com';
 55: EVTACT   node istype 'collapse';
 56: EVTEND   node istype 'collapse';

 57: EVTDONE  node  istype 'reg_D';
 58: HBADLAT  node  istype 'reg_D';
 59: BSYERR   pin  istype 'buffer';
 60: XDATA    pin  istype 'com,buffer';
 61: CLRWDT   pin  istype 'buffer';
 62: EECLR    pin  istype 'com,buffer';
 63: CLRBSY   pin  istype 'com,buffer';
 64: HEADDONE pin  istype 'reg,buffer';
 65: SER_EN   pin  istype 'com,buffer';
 66: EVTSTRT  pin  istype 'com,buffer';
 67: BSYERRL  node istype 'reg_D';
 68: HBAD     pin  istype 'com';
 69: EOFEVNT  pin  istype 'com';
 70: equations
 71:
 72: MRESET  = FE_CLR # !ABORTP;"Abortp polarity changed 1/29/98"
 73:
 74: " FE_CLR is the normal clear of busy and status flags after an event."
 75: " ABORT is a pulse received over the link if the event needs to be terminated"
 76: " for some reason. There is also a pulse on ABORT at powerup or on front"
 77: " panel reset to clear all status conditions."
 78:
 79: HEADCLK = ST00 & !ST01 & ADJSTRB & HPDAVT & !BSYERRL;"Clock for header counter"
 80: HEADCNT = HEADCLK;"Use as strobe pulse for ext. memory"
 81: HEADB15 = HQ0 & HQ1 & HQ2 & HQ3 & !HQ4 & !HQ5 & !HQ6; "Header word fifteen"
 82:
 83: EVTACT  =!ST00 & ST01 & LINKRDY & !BSYERRL;" start WDT and latch HEADDONE"
 84: HEADDONE.D = EVTACT;
 85: HEADDONE.CLK = ADJSTRB;
 86: HEADDONE.AR = MRESET;
 87: EVTSTRT = EVTACT;
 88: PEDCLK.D  = ST00  & HEADB15 & !TPCMODE;"PED_OFFSET_CLK used by SVT"
 89: PEDCLK.CLK = ADJSTRB & HPDAVT;
 90: EVTEND  = HPCAVT & HPDA02 & ADJSTRB & LINKRDY;
 91:
 92: "        Latch Event end State       "
 93: EVTDONE.D =!BSYERRL;
 94: EVTDONE.CLK = EVTEND;
 95: EVTDONE.AR  = MRESET;
 96: EOFEVNT = !EVTDONE;
 97: XDATA  = (ST00 & !ST01 & !TPCMODE);"mux low 10 bits to both asics(SVTMODE)"
 98: CLRWDT =!(EVTDONE # MRESET);
 99: EECLR  = !MRESET;
100: "Header Counter, used to select word 16 for PED_OFFSET_VALUE"
101: "                and check for exactly 64 words."
102: [HQ6..HQ0].CLK = HEADCLK;
103: [HQ6..HQ0].AR = MRESET # !LINKRDY;
104: HQ6.T = HQ5 & HQ4 & HQ3 & HQ2 & HQ1 & HQ0;
105: HQ5.T = HQ4 & HQ3 & HQ2 & HQ1 & HQ0;
106: HQ4.T = HQ3 & HQ2 & HQ1 & HQ0;
107: HQ3.T = HQ2 & HQ1 & HQ0;
108: HQ2.T = HQ1 & HQ0;
109: HQ1.T = HQ0;
110: HQ0.T = 1;
111:
112: " 'or' three busy signals for CHERST2 logic"

113: " busy is cleared when all signals are low"
114:
115: CLRBSY = CLRBSY0 # CLRBSY1 # CLRBSY2;
116:
117: "Enable VRAM for header information"
118: ST01D.D   = ST01; "Delay turnoff for last header word"
119: ST01D.CLK = ADJSTRB;
120: SEREN.CLK = ADJSTRB;
121: SEREN.S  = ST00;
122: SEREN.R  = ST01D;
123: !SER_EN  = SEREN;
124:
125: "Latch bad Header length"
126:  
127: HBADLAT.CLK = HPCAVT & HPDA01 & ADJSTRB & !BSYERRL;
128: HBADLAT.D   = !(HQ6 & !HQ5 & !HQ4 & !HQ3 & !HQ2 & !HQ1 & !HQ0);
129: HBADLAT.AR  = MRESET;
130: HBAD     = !(HBADLAT # !LINKRDY);
131:
132: "Latch a busy error if start of event is received while still busy"
133:
134: BSYERRL.CLK = HPCAVT & HPDA00 & ADJSTRB;
135: BSYERRL.D   = !BUSY;
136: BSYERRL.AR  = MRESET;
137: BSYERR    = !(BSYERRL # !LINKRDY);
138: end
139:


